

# Computing Signed Areas and Volumes with Maple

*Shengxiang Xia*<sup>1</sup>

e-mail: summer\_5069@yahoo.com.cn  
College of Science, Shandong Jianzhu University  
Jinan 250101, China

*Wei-Chi Yang*

e-mail: wyang@radford.edu  
Department of Mathematics and Statistics  
Radford University  
Radford, VA 24142, USA

*Vladimir V. Shelomovskii*

e-mail: vvsss@rambler.ru  
Murmansk State University  
Russia

## Abstract

*The signed area bounded by a parametric curve with respect to a slanted line and the signed volume bounded by a smooth parametric surface, a plane and the perpendicular projection from the surface to the plane were discussed in [1]. In this note, in addition to the integration method described in [1], we introduce a transformation method to further discuss how to find the volume of the solid bounded by an orientable smooth surface, especially for a real closed quadric surface and a plane. We present Maple procedures based on integration and transformation methods used to evaluate signed areas and volumes. The procedures are designed with formal parameters which can be easily used or modified by instructors and students.*

## 1. Introduction

The geometric significance of a Riemann integral  $\int_a^b f(x)dx$  is the net area of the region bounded by the curve  $y = f(x)$ , two vertical lines  $x = a$  and  $x = b$ , and the  $x$ -axis. The ways of finding the area bounded by a parametric curve  $[x(t), y(t)]$  ( $t_1 \leq t \leq t_2$ ), a slanted line  $y = mx + b$ , and the perpendiculars to the line from  $(x(t_1), y(t_1))$  to  $(x(t_2), y(t_2))$  in two dimensions were discussed in [1]. Since the graph of a function  $y = f(x)$  was replaced by a parametric curve, the sign of the bounded area depends on the direction of the given parametric curve. The formula computing signed areas was derived by a proper coordinate transformation (see [1]). In the 3-dimensional space  $R^3$ , when considering the volume bounded by a smooth parametric surface  $\mathbf{w}(s, t) = [x(s, t), y(s, t), z(s, t)]$ ,  $s_1 \leq s \leq s_2$ ,  $t_1 \leq t \leq t_2$ , with respect to a general plane, similar formulae computing the signed volumes were obtained in [1]. In this paper, Maple procedures based on these formulae are presented and their performance is demonstrated with illustrative examples. In section 3, we describe how we find the volume of the solid bounded by an orientable smooth surface (especially for a real closed quadric surface) and a plane with integral method and transformation method, and

---

<sup>1</sup> Supported by NSF of China (No. 51078225)

we use some examples to demonstrate when these two different methods will result in the same answers, and to show their respective characteristics.

## 2. Signed Areas

We first recall the following theorems from [1]:

**Theorem 2.1** ([1]) Let  $C$  be the smooth curve  $w(t) = [x(t), y(t)]$ , where  $t_1 \leq t \leq t_2$ . Let  $R$  be the region bounded by  $C$ , the line  $y = mx + b$ , and the perpendiculars to the line from  $(x(t_1), y(t_1))$  and  $(x(t_2), y(t_2))$ . Then the net signed area of  $R$  is given by

$$\frac{1}{1+m^2} \int_{t_1}^{t_2} (-x(t)m + y(t) - b)(x'(t) + y'(t)m) dt. \quad (1)$$

Throughout this paper, the word “area” represents a net signed area and similar terminology is used in the 3-dimensional space.

**Corollary 2.2** ([1]) Let  $C$  be a closed smooth curve traveling only in one direction, then (1) produces the area bounded by the closed curve which is independent of the location of the line.

Here is the Maple procedure:

```
>sarea:=proc(fun, m, b, t1, t2)
local graph, funt, ss, x1, y1, x2, y2;
# please use “t” as parameters in fun;
x1:=eval(fun[1], t=t1);
y1:=eval(fun[2], t=t1);
x2:=eval(fun[1], t=t2);
y2:=eval(fun[2], t=t2);
graph:=plot([[fun[1], fun[2], t=t1..t2], m*t+b, -1/m*(t-x1)+y1, -1/m*(t-x2)+y2], linestyle=[1, 2, 3, 4], scaling=constrained);
print(graph);
funt:=(-fun[1]*m+fun[2]-b)*(diff(fun[1], t)+diff(fun[2], t)*m);
ss:=int(funt, t=t1..t2);
ss:=ss/(1+m*m);
end;
```

We implement the Maple procedure above to the following Examples 2.3 and 2.4.

**Example 2.3** Let  $C$  be the curve of  $[8\sin(3t)\cos(t), 8\sin(3t)\sin(t)]$ , where  $t \in [0, \pi/2]$ , the line  $L$  be  $y = -x - 9$ . Find the area bounded by  $C$  with respect to the line  $L$ .

**Solution:** Set up the expression for  $C$ , then run the procedure “sarea”, we get the following result.

>fun:=[8\*sin(3\*t)\*cos(t), 8\*sin(3\*t)\*sin(t)];

>S:=sarea(fun, -1, -9, 0, Pi/2);

$$S := -8\pi + 20$$




Figure 1. Area bounded by  $C$  with respect to  $L$  (green line)

We note that both yellow and blue lines in Figure 1 are perpendicular to the green line segment. We remark that the answer  $-8\pi + 20$  is the net signed area, it is not the actual area of the region. The region consists of two portions:  $S_1$  is the loop going from origin back to itself in a counterclockwise direction and the other  $S_2$  is enclosed by an arc and three segments in a clockwise direction. We demonstrate the areas of  $S_1$  and  $S_2$  respectively below.

>S1:=sarea(fun, -1, -9, 0, Pi/3);

$$S1 := -\frac{16}{3}\pi$$

We note that even if there is a straight line (yellow line) cuts through  $S_1$  but  $S_1$  is a closed curve traveling only in one direction, it follows from Corollary 2.2 that calculation of the area of  $S_1$  is not affected even if the perpendiculars intersect the curve  $C$ . The area of  $S_2$  is shown below:

>S2:=sarea(fun, -1, -9, Pi/3, Pi/2);

$$S2 := -\frac{8}{3}\pi + 20$$

The true real area bounded by  $C$  respect to  $L$  is  $|S_1| + |S_2| = \frac{8}{3}\pi + 20$ .

**Example 2.4** Let  $C$  be the curve  $[9\sin(3t)\cos(t), 7\sin(3t)\sin(t)]$ , where  $t \in [\frac{\pi}{3}, \frac{\pi}{2}]$ , the line  $L$  be  $y = -x - 5$ . Find the area bounded by  $C$  with respect to the line  $L$  (see Figure 2).

**Solution:** Set up the expression for  $C$ , then run the procedure “sarea”, we obtain the following result.

```
>fun1:=[9*sin(3*t)*cos(t), 7*sin(3*t)*sin(t)];
>sarea(fun1, -1, -5, Pi/3, Pi/2);
```

$$\frac{21}{4} - \frac{21}{8}\pi$$

Note that the answer  $\frac{21}{4} - \frac{21}{8}\pi$  is the net signed area, it is not the real area of the region. By Figure 2,

we know that the line  $L$  and the curve  $C$  have one intersection point  $P$ , the two regions ( $S_1$  and  $S_2$ ) are located on the opposite sides of the line  $L$ , the signed areas of the regions have different signs. Hence, to get the real area of the region we carry out the following calculations. Firstly, we find the  $t$ -value corresponding to the intersection point  $P$ . Solving the follow equation: evalf(solve ('7\* sin(3\*t)\*sin(t)= -9\*sin(3\*t)\*cos(t)-5', t), 15), we obtain  $t_0= 1.22964231411985$ . Then we use the procedure on  $[\frac{\pi}{3}, t_0]$  and  $[t_0, \frac{\pi}{2}]$  respectively, we obtain  $S_1=1.791900156$ ,  $S_2=-4.788580872$ . Therefore the real area is  $|S_1|+|S_2|= 6.580481028$ .

```
>S1:=sarea(fun1, -1, -5, Pi/3, 1.22964231411985);
S1:= 1.791900156
>S2:=sarea(fun1, -1, -5, 1.22964231411985, Pi/2);
S2:= -4.788580872
```




Figure 2. Area bounded by  $C$  with respect to  $L$

We verify here that the signed areas for  $S_1$  and  $S_2$  are consistent with the Green's Theorem students encountered in a multi-variable calculus class. Let  $C_1$  be the path traversed counterclockwise from  $\overline{OM}$  to  $\overline{MP}$  to  $PO$  and  $C_2$  be the path traversed clockwise from  $\overline{PN}$  to  $\overline{NQ}$  to  $QP$ . Using Green's Theorem on the curve  $C_1$ , we obtained 1.791900155997022 for the area of the first region  $S_1$ . Using Green's Theorem on the curve  $C_2$ , we obtained -4.788580871670222 for the area of the second region  $S_2$ .

$$S_1 = \frac{1}{2} \int_{C_1} (x(t)y'(t) - y(t)x'(t))dt = 1.791900155997022. \quad S_2 = \frac{1}{2} \int_{C_2} (x(t)y'(t) - y(t)x'(t))dt = -4.788580871670222.$$

### 3. Signed Volumes

We recall the following theorem from [1] when dealing with a volume with respect to a slanted plane:

**Theorem 3.1.** Let  $S$  be the smooth parametric surface  $w(s,t)=[x(s,t),y(s,t),z(s,t)]$ ,  $s_1 \leq s \leq s_2$ ,  $t_1 \leq t \leq t_2$ . Let  $R$  be the region bounded by  $S$ , the plane  $P:ax+by+cz=d$ , and the perpendicular projection from the surface to the plane  $P$ . Then the net signed volume of  $R$  is given by

$$\iint r(s,t)dpdq = \int_{s_1}^{s_2} \int_{t_1}^{t_2} r(s,t) \begin{vmatrix} \frac{\partial p}{\partial s} & \frac{\partial p}{\partial t} \\ \frac{\partial q}{\partial s} & \frac{\partial q}{\partial t} \end{vmatrix} dt ds, \quad (2)$$

where  $\begin{pmatrix} p(s,t) \\ q(s,t) \\ r(s,t) \end{pmatrix} = [p_1, p_2, p_1 \times p_2]^{-1} \begin{pmatrix} x(s,t) \\ y(s,t) \\ z(s,t) - \frac{d}{c} \end{pmatrix}$ , and  $p_1, p_2$  are two orthonormal vectors on the

plane  $P$ .

**Corollary 3.2** ([1]). If the smooth surface  $w(s,t)$  is an orientable closed surface, the equation (2) produces the volume of the region bounded by the closed surface, and this volume is independent of the location of the plane  $P$ .

Here is the Maple procedure for Theorem 3.1:

```
>svolume:=proc(fun, PL, ST)
local a1, b1, c1, d1, p1, p2, p3, n1, n2, n3, dotp, mat, wt, jac, tjac, ss, fvec, pars;
# please use "s, t" (or other variables but in alphabet) as parameters in fun;
# PL=[a, b, c, d] be the coefficients of the plane "ax+by+cz=d";
# ST=[s1, s2, t1, t2] be bounds of the integrals;
a1:=PL[1]; b1:=PL[2]; c1:=PL[3]; d1:=PL[4];
if c1=0 then
    error ("the case can not be dealt with");
end if;
if d1<>0 and a1<>0 and b1<>0 then
    p1:=[0, 0, d1/c1]; p2:=[d1/a1, 0, 0]; p3:=[0, d1/b1, 0];
elif d1<>0 and a1=0 and b1<>0 then
    p1:=[0, 0, d1/c1]; p2:=[0, d1/b1, 0]; p3:=[1, 0, d1/c1];
elif d1<>0 and a1<>0 and b1=0 then
    p1:=[0, 0, d1/c1]; p2:=[d1/a1, 0, 0]; p3:=[0, 1, d1/c1];
elif d1<>0 and a1=0 and b1=0 then
    p1:=[0, 0, d1/c1]; p2:=[0, 1, d1/c1]; p3:=[1, 0, d1/c1];
```

```

elif d1=0 and a1<>0 and b1<>0 then
    p1:=[0, 0, 0]; p2:=[1, 0, -a1/c1]; p3:=[0, 1, -b1/c1];
elif d1=0 and a1=0 and b1<>0 then
    p1:=[0, 0, 0]; p2:=[1, 1, -b1/c1]; p3:=[2, 1, -b1/c1];
elif d1=0 and a1<>0 and b1=0 then
    p1:=[0, 0, 0]; p2:=[1, 1, -a1/c1]; p3:=[1, 2, -a1/c1];
elif d1=0 and a1=0 and b1=0 then
    p1:=[0, 0, 0]; p2:=[1, 0, 0]; p3:=[0, 1, 0];
end if;
n1:=vector(p2-p1); n2:=vector(p3-p1);
dotp:=linalg[dotprod](n1, n2);
if dotp<>0 then
    n1:=evalm(n1-dotp*n2/(linalg[norm](n2, 2))^2) ;
end if;
n1:=evalm(n1/linalg[norm](n1, 2)); n2:=evalm(n2/linalg[norm](n2, 2));
n3:=linalg[crossprod](n1, n2);
mat:=matrix(3, 3, [n1, n2, n3]); # mat is an orthogonal matrix;
mat:=linalg[transpose](mat); mat:=linalg[inverse](mat);
fvec:=vector( [fun[1], fun[2], fun[3]-d1/c1]); wt:=linalg[multiply](mat, fvec);
jac:=MTM[jacobian]([wt[1], wt[2]], [s, t]); tjac:=linalg[det](jac);
ss:=int(int(wt[3]*tjac, t=ST[3]..ST[4]), s=ST[1]..ST[2]);
end:

```

We implement the above 3D Maple procedure to show in next Example 3.3 that the volume of the closed ellipsoid is independent of the location of the plane  $P$  (see Corollary 3.2).

**Example 3.3** Let  $S$  be the ellipsoid  $\begin{bmatrix} x(s,t) \\ y(s,t) \\ z(s,t) \end{bmatrix} = \begin{bmatrix} -1+5\sin t \cos s \\ 2+3\sin t \sin s \\ -3+8\cos t \end{bmatrix}$ , where  $s \in [0, 2\pi]$ , and  $t \in [0, \pi]$ ,

and let  $P$  and  $P_1$  be the planes  $3x - 2y + 5z = 30$  and  $3x + 2y - 3z = -5$  respectively. Find the volume of the region  $R$  bounded by  $S$ , the plane and the perpendicular projection from the surface to the plane  $P$ . Repeat the same process to find the volume of the region  $R$  if we replace  $P$  by  $P_1$ .

**Solution:** We first set up the expression for ellipsoid  $S$ , draw the figures of  $S$  and the plane  $P$ , which does not intersect  $S$  (see Figure 3). We then run the Maple procedure “**svolume**”, to obtain the signed volume of  $S$  to be  $-160\pi$ . We demonstrate that the volume of the closed ellipsoid is independent of the location of the plane  $P$  (from Corollary 3.2). We use plane  $P_1$  (see Figure 4), which intersects  $S$ , to demonstrate this effect.

```

>fun:=[-1+5*sin(t)*cos(s), 2+3*(sin(t)*sin(s)), -3+8*cos(t)];
>graph1:= plot3d(fun, s = 0 .. 2*Pi, t = 0 .. Pi, axes = boxed);
>graph2:= implicitplot3d(3*x-2*y+5*z = 30, x=-8 ..8, y=-8.. 8, z=-8 ..8);
>display({graph1, graph2});
>V1:=svolume(fun, [3, -2, 5, 30], [0, 2*Pi, 0, Pi]);

```

$$V1 := -160\pi$$

```
>graph3:=implicitplot3d(3*x+2*y-3*z=-5, x=-8..8, y=-8..8, z=-8..8);
>display({graph1, graph3});
>V2:=svolume(fun, [3, 2, -3, -5], [0, 2*Pi, 0, Pi]);
V2 := -160\pi
```




Figure 3. Volume of  $S$  with respect to  $P$




Figure 4. Volume of  $S$  with respect to  $P_1$

In the next example, we use Maple procedure “**svolume**” to find the volume of the region bounded by an open surface and a plane, and the perpendicular projection from the surface to the plane.

**Example 3.4** Let  $S$  be the ellipsoid

$$\begin{bmatrix} x(s,t) \\ y(s,t) \\ z(s,t) \end{bmatrix} = \begin{bmatrix} \cos(4) \sin t \cos s - 3 \sin(4) \cos t \\ \sin(2) \sin(4) \sin t \cos s + 2 \cos(2) \sin t \sin s + 3 \sin(2) \cos(4) \cos t \\ \cos(2) \sin(4) \sin t \cos s - 2 \sin(2) \sin t \sin s + 3 \cos(2) \cos(4) \cos t \end{bmatrix},$$

where  $s \in [\pi, 2\pi]$ , and  $t \in [0, \pi/2]$ , and  $P$  be the plane  $x + y + z = -3$ . Find the volume of the region bounded by  $S$ , the plane  $P$ , and the perpendicular projection from the surface  $S$  to the plane  $P$ .

**Solution:** We set up the expression for  $S$ , and plot the figure of  $S$  and  $P$ . Then we run the procedure “**svolume**”, and get the real volume is -15.55764889.

```
>fun2:=[cos(4)*sin(t)*cos(s)-3*sin(4)*cos(t), sin(2)*sin(4)*sin(t)*cos(s)+ 2*cos(2)*sin(t)*sin(s)+
3*sin(2)*cos(4)*cos(t), cos(2)*sin(4)*sin(t)*cos(s)-2*sin(2)*sin(t)*sin(s)+3*cos(2)*cos(4)*cos(t)];
>graph4:=plot3d(fun2, s = Pi.. 2*Pi, t = 0..Pi/2, axes=boxed);
>graph5:= implicitplot3d(x+y+z=-3, x =-3..3, y=-3..3, z =-5..5);
>display(graph4, graph5);
>V1:=svolume(fun2, [1, 1, 1, -3], [ Pi, 2*Pi, 0, Pi/2]);
>evalf(V1);
-15.55764889
```




Figure 5. Volume of a surface  $S$  with respect to the plane  $P$

**Remark:** In Theorem 3.1, if  $S$  and  $P$  have no intersection (i.e.  $S$  is located in one side of  $P$ ), then the real volume of the region  $R$  can be easily calculated by (2), it is absolute value of the number obtained by (2). But for a general case, it is difficult to find the real volume of the region  $R$ , unless the surface  $S$  is a closed surface (see Example 3.3).

### 3.1. Integral Method

We now turn our attention to find part of the volume for a smooth orientable surface when a plane cuts through the given surface. In particular, suppose that a plane  $P$  cuts the solid bounded by the surface  $S$  of the form  $F(x, y, z) = 0$  into only two pieces. In this section, we discuss the following Theorem of finding the volumes of these two respective pieces bounded by  $P$  and  $S$  mentioned in [1].

**Theorem 3.5** [1]. Let  $S$  be the smooth parametric surface of the form  $F(x, y, z) = 0$ . Let  $R$  be the region bounded by  $S$  and the plane  $P: ax + by + cz = d$ . Furthermore, we assume the following conditions are met:

(i) If  $G(p, q, w) = F \left( A \begin{bmatrix} p \\ q \\ w \end{bmatrix} + \begin{bmatrix} 0 \\ 0 \\ d/c \end{bmatrix} \right) = 0$ , where  $A = [p_1, p_2, n]$  and  $p_1, p_2$  are two

perpendicular unit vectors on the plane  $P$ , and  $n$  is the unit normal vectors of  $P$  that points toward the region whose volume we seek.

(ii)

(ii) If  $G(\rho \sin t \cos s, \rho \sin t \sin s, \rho \cos t) = 0$  is solvable for  $\rho$ ,  $\rho = \rho(s, t)$  is a real function and represent one piece of  $S$ , then the volume of  $R$  is given by

$$\int_0^{2\pi} \int_0^{\frac{\pi}{2}} \left( \int_0^{\rho(s,t)} (r^2 \sin t) dr \right) dt ds. \quad (3)$$

We call above method as **the integral method**, and we make the steps of the detailed calculations into the following Maple procedure.



## The Maple procedure

```

>parvol:=proc(fun, PL, neworigin)
local a1, b1, c1, d1, p1, p2, p3, n1, n2, n3, dotp, mat, svol, svol2, fvec, xnew, equ, equ1, rou, fun1,
fun2, para, valst, vals, norigin;
# please use "x, y, z" (or other variables but in alphabet) as parameters in fun;
# PL=[a, b, c, d] be the coefficients of the plane "ax+by+cz=d";
# neworigin is the origin of a new coordinate system, the given value is neworigin=[0, 0, d/c];
a1:=PL[1]; b1:=PL[2]; c1:=PL[3]; d1:=PL[4];
if nargs<3 then
    norigin:=[0, 0, d1/c1];
else
    norigin:=neworigin;
end if;
para:=MTM[findsym](fun);
if c1=0 then
    error ("the case can not be dealt with");
end if;
if d1<>0 and a1<>0 and b1<>0 then
    p1:= [0, 0, d1/c1]; p2:= [d1/a1, 0, 0]; p3:= [0, d1/b1, 0];
elif d1<>0 and a1=0 and b1<>0 then
    p1:= [0, 0, d1/c1]; p2:= [0, d1/b1, 0]; p3:= [1, 0, d1/c1];
elif d1<>0 and a1<>0 and b1=0 then
    p1:= [0, 0, d1/c1]; p2:= [d1/a1, 0, 0]; p3:= [0, 1, d1/c1];
elif d1<>0 and a1=0 and b1=0 then
    p1:= [0, 0, d1/c1]; p2:= [0, 1, d1/c1]; p3:= [1, 0, d1/c1];
elif d1=0 and a1<>0 and b1<>0 then
    p1:= [0, 0, 0]; p2:= [1, 0, -a1/c1]; p3:= [0, 1, -b1/c1];
elif d1=0 and a1=0 and b1<>0 then
    p1:= [0, 0, 0]; p2:= [1, 1, -b1/c1]; p3:= [2, 1, -b1/c1];
elif d1=0 and a1<>0 and b1=0 then
    p1:= [0, 0, 0]; p2:= [1, 1, -a1/c1]; p3:= [1, 2, -a1/c1];
elif d1=0 and a1=0 and b1=0 then
    p1:= [0, 0, 0]; p2:= [1, 0, 0]; p3:= [0, 1, 0];
end if;
n1:=vector(p2-p1); n2:=vector(p3-p1); dotp:=linalg[dotprod](n1, n2);
if dotp<>0 then
    n1:=evalm(n1-dotp*n2/(linalg[norm](n2, 2))^2);
end if;
n1:=evalm(n1/linalg[norm](n1, 2)); n2:=evalm(n2/linalg[norm](n2, 2));
n3:=linalg[crossprod](n1, n2);
mat:=matrix(3, 3, [n1, n2, n3]); mat:=linalg[transpose](mat);
fvec:=matrix( 3,1, [p, q, r]);
xnew:=evalm((mat*fvec)+matrix(3,1, norigin));
equ:=MTM[subst](fun=0, [para[1], para[2], para[3]], [xnew[1, 1], xnew[2,1], xnew[3, 1]]);

```

```

equ1:=MTM[sub](equ, [p, q, r], [d*sin(t)*cos(s), d*sin(t)*sin(s), d*cos(t)]);
rou:=solve(equ1, d);
fun1:=simplify(rou[1]); fun2:=simplify(rou[2]);
valst:=int(v^2*sin(t), v=0..fun1);
vals:=evalf(student[simpson](valst(s, t), t=0..Pi/2, 200));
svol:=evalf(student[simpson](vals, s=0..2*Pi, 200));
print("the volume of one piece bounded by the plane P and the surface S is");
print(svol);
if (fun2 <> null) then
    valst:=int(v^2*sin(t), v=0..fun2); vals:=evalf(student[simpson](valst, t=0..Pi/2, 200));
    svol2:=evalf(student[simpson](vals, s=0..2*Pi, 200));
    print("the volume of the other piece bounded by the plane P and the surface S is");
    print(svol2);
end if;
end:

```

**Note:** As mentioned in Theorem 3.5 of [1] that we need to first make sure that  $\rho$  can be solved by Maple or our Maple procedure can be applied to obtain the result. In addition, it is not easy to get the exact integration value from (3). We sometimes need to apply higher dimension Simpson's rule or integration quadratures to obtain an approximation for the integrals in the procedure.

In the following two examples 3.6 and 3.7 when  $S$  is an elliptic paraboloid or a hyperboloid of two sheets, we show how we transform the surface equation to a standard form before applying our Maple procedure.

**Example 3.6** Let  $S$  be the surface  $2x^2 + 2y^2 + 3z^2 + 4xy + 2xz + 2yz - 4x + 6y - 2z + 3 = 0$  and the plane  $P$  be  $\sqrt{2}x - \sqrt{2}y + \sqrt{3}z - 20 = 0$ . Find the volume of the region bounded by  $S$  and  $P$ .

**Solution:** Set up the expression for  $S$ , draw the figure of  $S$  ( $S$  is an elliptic paraboloid, see Figure 6) and  $P$ . If we directly run the procedure “**parvol**”, we can't get the correct answer immediately.

```

>with(plots):
>fun:= 2*x^2+2*y^2+3*z^2+4*x*y+2*x*z+2*y*z-4*x+6*y-2*z+3:
>fig1:=implicitplot3d({ fun}, x=-2..15, y=-15..2, z=-7..10, numpoints=2000, axes=boxed):
>fig2:=plot3d({-(sqrt(2)*x-sqrt(2)*y-20)/sqrt(3)}, x=-2..15, y=-15..2, shading=zhue, transparency
=0.3, numpoints=2000, axes=boxed):
>display(fig1, fig2);

```

Let us convert the quadric  $S$  into a standard equation by the orthogonal transformation:

$$\begin{cases} x = \frac{\sqrt{3}}{3}x' + \frac{\sqrt{6}}{6}y' - \frac{\sqrt{2}}{2}z' - \frac{1}{4} \\ y = \frac{\sqrt{3}}{3}x' + \frac{\sqrt{6}}{6}y' + \frac{\sqrt{2}}{2}z' - \frac{1}{4} \\ z = \frac{\sqrt{3}}{3}x' - \frac{\sqrt{6}}{6}y' - \frac{\sqrt{2}}{2}z' + \frac{1}{2} \end{cases}$$

Then the equations of  $S$  and  $P$  under the new coordinate system  $x'y'z'$  are  $S_1: 5x'^2 + 2y'^2 + 5\sqrt{2}z' + \frac{9}{4} = 0$ , and  $P_1: x' - \sqrt{2}y' - 2z' + \frac{\sqrt{3}}{2} - 20 = 0$  respectively. Hence the volume bounded by the surface  $S$  and the plane  $P$  equals the volume bounded by  $S_1$  and  $P_1$ . We run the procedure “**parvol**” for equations of  $S_1$  and  $P_1$ , we get the volume to be -335.8952501.

```
>fun1:=5*x^2+2*y^2+5*sqrt(2)*z+9/4;
>fig3:= implicitplot3d({fun1}, x=-5..5, y=-8..8, z=-16..1, numpoints =2000, axes=boxed);
>fig4:=plot3d({(x-sqrt(2)*y+sqrt(3)/2-20)/2}, x=-5..5, y=-8..8, shading=zhue, transparency=0.3, numpoints=2000, axes=boxed);
>display(fig3, fig4);
```




Figure 6. The region bounded by  $S$  and  $P$




Figure 7. The region bounded by  $S_1$  and  $P_1$

```
>parvol(fun1, [1, -sqrt(2), -2, 20-sqrt(3)/2]);
    “the volume of one piece bounded by the plane P and the surface S is”
    1.170009574 1010
    “the volume of the other piece bounded by the plane P and the surface S is”
    -335.8952501
```

**Note:** If we verify  $n_3$  (the meaning as  $n$  in Theorem 3.5) in the procedure “**parvol**”, we see that its direction is downward, thus the negative value is the volume we seek.

We can find that the projection of intersection of  $S_1$  and  $P_1$  on the  $xy$ -plane is  $5x^2 + 2y^2 + \frac{5}{2}\sqrt{2}x - 5y + \frac{19}{4} - 5\sqrt{2} = 0, z = 0$ . Thus, we may find the volume bounded by  $S_1$  and  $P_1$  the following integral.

```
>V:=evalf(int(int(1/2*(x-sqrt(2)*y+sqrt(3)/2-20)+1/(5*sqrt(2))*(5*x^2+2*y^2+9/4),
y=5/4-(1/4)*sqrt(-13-20*sqrt(2)*x+400*sqrt(2)-40*x^2)..5/4+(1/4)*sqrt(-13-20*sqrt(2)*x+400*sqrt
(2)-40*x^2)), x=-(1/4)*sqrt(2)-(1/5)*sqrt(-5+250*sqrt(2))..-(1/4)*sqrt(2)+(1/5)*sqrt(-5+250
*sqrt(2))));
V:= -335.8691901
```

For this Example 3.6, we point out our quadrature yields the same answer as the one by using traditional method if we know explicitly what the intersection is and it is easy to use  $dx dy$  or  $dy dx$  integrations.

In the following example, we consider the case where  $S$  is a hyperboloid of two sheets.

**Example 3.7** Let  $S$  be the surface  $x^2 + y^2 + z^2 + 4xy - 4xz - 4yz - 5 = 0$  and  $P$  be  $\sqrt{6}x + \sqrt{3}y - \sqrt{6}z + 21 = 0$ . Find the volume of the region bounded by  $S$  and  $P$ .

**Solution:** Set up the expression for  $S$ , draw the figure of  $S$  ( $S$  is a hyperboloid of two sheets, see Figure 8) and  $P$ . If we run the procedure “**parvol**” directly, we won’t get the correct answer immediately.

```
>with(plots):
>fun:= x^2+y^2+z^2+4*x*y-4*x*z-4*y*z-5:
>fig5:=implicitplot3d({fun}, x=-30..20, y=-25..20, z=-20..30, numpoints =2000, axes=boxed):
>fig6:=plot3d({(sqrt(6)*x+sqrt(3)*y+21)/sqrt(6)}, x=-30..20, y=-25..20, shading=zhue,
transparency=0.3, numpoints=2000, axes=boxed):
> display(fig5, fig6);
```

Instead, we first convert the quadric  $S$  into a standard equation by the orthogonal transformation:

$$\begin{cases} x = \frac{\sqrt{3}}{3}x' + \frac{\sqrt{2}}{2}y' + \frac{\sqrt{6}}{6}z' \\ y = -\frac{\sqrt{3}}{3}x' - \frac{\sqrt{6}}{6}z' \\ z = -\frac{\sqrt{3}}{3}x' + \frac{\sqrt{2}}{2}y' - \frac{\sqrt{6}}{6}z' \end{cases}$$

Then the equations of  $S$  and  $P$  in the new coordinate system  $x'y'z'$  are  $S_1 : 5x'^2 - y'^2 - z'^2 - 5 = 0$ , and  $P_1 : (1+2\sqrt{2})x' + (2-\sqrt{2})z' + 21 = 0$  respectively. Hence the volume bounded by the surface  $S$  and the plane  $P$  equals the volume bounded by  $S_1$  and  $P_1$ . We run the procedure “**parvol**” for equations of  $S_1$  and  $P_1$ , we obtain the volume to be -960.3695172.

```
>fun1:=5*x^2-y^2- z^2-5:
>fig7:= implicitplot3d({fun1}, x=-10..10, y=-15..15, z=-20..20, numpoints =2000, axes=boxed):
>fig8:=implicitplot3d({(1+2*sqrt(2))*x+(2-sqrt(2))*z+21=0}, x=-15..15, y=-15..15, z=-20..20,
shading=zhue, transparency=0.3, numpoints=2000, axes=boxed);
>display(fig7, fig8);
```





Figure 8. The region bounded by  $S$  and  $P$       Figure 9. The region bounded by  $S_1$  and  $P_1$

```
>parvol(fun1, [1+2*sqrt(2), 0, 2-sqrt(2), -21], [-21/(1+2*sqrt(2)), 0, 0]);
“the volume of one piece bounded by the plane P and the surface S is”
-960.3695172
“the volume of the other piece bounded by the plane P and the surface S is”
1.813683906 1010
```

**Note:** If we examine the normal vector  $n_3$  in the procedure “**parvol**”, we see that the direction of  $n_3$  is pointed toward the right, thus the net signed volume of -960.3695172 is the volume we seek, and we discard the second Maple output. As we did in Example 3.6, we may calculate the volume of the region bounded by  $S_1$  and  $P_1$  by using traditional method and get  $V= 960.3693067$ .

In view of Examples 3.6 and 3.7, we see that the equation of  $S$  needs to be converted into a standard form through an orthogonal transformation before using the procedure “**parvol**”. The Example 3.8 shows that this step is not necessary for certain ellipsoids. However, some special care needs to be addressed before the procedure “**parvol**” is used, see Example 3.9.

**Example 3.8** Let  $S$  be the surface  $14x^2 + 27y^2 + 98z^2 + 4yz - 34xz - 8xy + 28x + 20y - 74z - 169 = 0$

and  $P$  be the plane  $x + 3y - 2z = 1$ . Find the volumes of the regions bounded by the surface and the plane.

**Solution:** Set up the expression for  $S$ , draw the figure of  $S$  ( $S$  is an ellipsoid, see Figure 10) and  $P$ . Then we run the procedure “**parvol**”, we get the volumes are 53.39430479 and -14.90627780 respectively.

```
>fun1:=14*x^2+27*y^2+98*z^2+4*y*z-34*x*z-8*x*y+28*x+20*y-74*z-169:
>fig9:= implicitplot3d({fun1}, x=-5..5, y=-4..4, z=-3..3, numpoints=1000, axes=boxed):
>fig10:=implicitplot3d({x+3*y-2*z-1=0}, x=-6..5, y=-4..4, z=-2..2, shading=zhue, transparency
=0.3, numpoints=1000, axes=boxed):
>display(fig9, fig10);
>parvol(fun1, [1, 3, -2, 1]);
“the volume of one piece bounded by the plane P and the surface S is”
```

53.39430479  
 “the volume of the other piece bounded by the plane P and the surface S is”  
 -14.90627780




Figure 10. Volume bounded by  $S$  and  $P$

**Note:** The total volume, by taking the absolute value, is about 68.30058259. The surface  $S$  of example 3.8 is an ellipsoid which has a volume of  $\frac{1}{167} \cdot \frac{4}{3} \pi \cdot (195)^{\frac{3}{2}} \approx 68.30052710$ .

In the next example, we show that we need to take special care before adopting the integration method mentioned in Theorem 3.5. Without loss of generality, we select the surface that is already in standard form.

$$x^2 + \frac{y^2}{9} + \frac{(z-3)^2}{16} - 1 = 0$$

**Example 3.9** Let  $S$  be the surface  $x^2 + \frac{y^2}{9} + \frac{(z-3)^2}{16} - 1 = 0$ , and  $P$  be the plane  $3x - 2y - z = -1$ . Find the respective volumes bounded by the ellipsoid and the plane.

**Solution:** Set up the expression for  $S$ , draw the figures of  $S$  and  $P$  (see Figure 11). Then run the

procedure “**parvol**”, we get the volumes are 15.71173760 and -34.57548356 respectively.

```
>fun2:=x^2+y^2/9+(z-3)^2/16-1:
>with(plots):
>fig11:=implicitplot3d({fun2}, x=-2..2, y=-3..3, z=-2..8, numpoints =2000, axes=boxed):
>fig12:=implicitplot3d({3*x-2*y-z+1=0}, x=-2..2, y=-3..3, z=-2..8, shading=zhue, transparency
=0.3, numpoints=2000, axes=boxed):
>display(fig11, fig12);
>parvol(fun2, [3, -2, -1,-1]);
    “the volume of one piece bounded by the plane P and the surface S is”
    15.71173760
    “the volume of the other piece bounded by the plane P and the surface S is”
    -34.57548356
```




Figure 11. Volume bounded by  $S$  and  $P$

However, if we use the plane  $P_1: 5x - 3y - z = -12$ , we get the following wrong results, and they are complex numbers.

```
>parvol(fun2, [5, -3, -1,-12]);
    “the volume of one piece bounded by the plane P and the surface S is”
    53.48787193-3.016654431 I
    “the volume of the other piece bounded by the plane P and the surface S is”
    -1.169514883+0.03878192866 I
```

In the procedure “**parvol**”, we note that we use  $[0, 0, d/c]=[0, 0, 12]$  as the origin; however, we find that the point  $(0, 0, 12)$  is not in the region bounded by  $S$ . To correct this problem, *the new origin should be chosen as a point that is on the plane  $P_1$  and in the interior of the region bounded by  $S$* . For example, if we choose the point  $(-0.8, 5/3, 3)$  as the new origin, then we obtain the



following right answer, the verification will be shown in Example 3.12.

```
>parvol(fun2, [5,-3,-1,-12], [-0.8, 5/3, 3]);
    “the volume of one piece bounded by the plane P and the surface S is”
    49.04701500
    “the volume of the other piece bounded by the plane P and the surface S is”
    -1.211028031
```

Similarly, if we take the plane  $P_2: 3x - y + z = -1$  into consideration, then we again have following wrong results when the procedure “**parvol**” is used directly.

```
>fig14:=implicitplot3d({3*x-y+z=-1}, x=-2..2, y=-3..4, z=-2..8, shading=zhue, transparency=0.3,
numpoints=2000, axes=boxed);
>display(fig11, fig14);
>parvol(fun2, [3, -1, 1,-1]);
    “the volume of one piece bounded by the plane P and the surface S is”
    0
    “the volume of the other piece bounded by the plane P and the surface S is”
    -43.60938130
```




Figure 12. Volume bounded by  $S$  and  $P_2$

In this case, we note that the origin is  $[0, 0, d/c] = [0, 0, -1]$ , and the point  $(0, 0, -1)$  is on the boundary of the region bounded by  $S$ . If we choose the point  $(0, 2, 1)$ , which is on the plane  $P_2$  and inside  $S$ , as the new origin, then we get the right answer as shown below when we run the procedure “**parvol**”; we may verify the answer later in Example 3.12.

```
>parvol(fun2, [3, -1, 1,-1], [0, 2, 1]);
    “the volume of one piece bounded by the plane P and the surface S is”
    3.328215450
```



“the volume of the other piece bounded by the plane P and the surface S is”  
-46.93743190

In view of Theorem 3.5, the difficulty of finding  $\rho$  may prevent us getting the right answer. In Example 3.9, we demonstrated that the trick is the new origin should be correctly chosen, inside the cross section of the intersection, before applying the integration method in theorem 3.5.

Now, we introduce the following alternative transformation method.

### 3.2. Transformation Method

Suppose the surface  $S$  is a real closed quadric of the form:

$$F(x, y, z) = X^T AX + 2BX + C = 0, \quad X = \begin{pmatrix} x \\ y \\ z \end{pmatrix}, \quad A = \begin{pmatrix} a_{11} & a_{12} & a_{13} \\ a_{12} & a_{22} & a_{23} \\ a_{13} & a_{23} & a_{33} \end{pmatrix}, \quad B = (a_{14}, a_{24}, a_{34}), \quad C = a_{44}. \quad (4)$$

And the plane  $P: ax + by + cz = d$  cuts the solid bounded by the surface  $S$  into only two pieces. In this section, we discuss how we can use linear transformations to find the volumes of these two respective pieces bounded by  $P$  and  $S$ . We briefly outline our strategy of finding the desired volume as follows. According to the classifications of quadrics,  $S$  can be transformed into an ellipsoid  $S'$  of the form  $\frac{x^2}{a^2} + \frac{y^2}{b^2} + \frac{z^2}{c^2} = 1$  by a proper orthogonal transformation (including moving and rotation)

[4], and  $P$  is transformed into a new plane  $P_1$  correspondingly. Furthermore, we adopt a linear transformation:  $x = ax', y = by', z = cz'$  so that  $S'$  and the plane  $P_1$  are mapped into a unit sphere  $S''$  and a new plane  $P_2: a_2x' + b_2y' + c_2z' = d_2$  respectively. Consequently, we note that finding the volume of one piece bounded by  $P$  and  $S$  is transformed into finding the volume bounded by a spherical segment by  $P_2$  and  $S''$ . It is known that the volume of a spherical segment is  $\pi h^2(R - h/3)$ , where  $R$  is the radius of the sphere and  $h$  is the height of the segment. Suppose we let  $d$  be the distance from the origin to the plane  $P_2$ , then  $h = R - d$  (or  $h = R + d$ ). By the linearity of the transformation, the ratio of the volumes of the parts of the sphere  $S''$  and  $S$  should be the same. Finally, with the Jacobian determinant determined by the linear transformation, we will obtain the exact volumes of the two pieces bounded by  $P$  and  $S$ .

The main computing steps are as follows: Firstly, we determine if the  $S$  is a real closed quadric (i.e. an ellipsoid) (see Step 1). Secondly, finding the principal diametral planes of  $S$ , we make the orthogonal transformation by using the three orthogonal principal diametral planes as new coordinate planes, then  $S$  is transformed into  $S'$  of the form  $\frac{x^2}{a^2} + \frac{y^2}{b^2} + \frac{z^2}{c^2} = 1$  (see Steps 2 through 6 below). Next, we make use of a linear transformation such that those two respective pieces bounded by  $P$  and  $S$  are changed into spherical segments. Finally, the volumes of two respective pieces bounded by  $P$  and  $S$  are obtained (see Steps 7 through 8).

**Step 1.** We compute the invariants (under orthogonal transformations) of  $F(x, y, z) = 0$ ,  $I_1, I_2, I_3$  and  $I_4$ , to determine whether the  $S$  is a real closed quadric (i.e. an ellipsoid) by checking the

conditions:  $I_2 > 0$ ,  $I_1 I_3 > 0$ , and  $I_4 < 0$ . Where

$$I_1 = a_{11} + a_{22} + a_{33}, I_2 = \begin{vmatrix} a_{11} & a_{12} \\ a_{12} & a_{22} \end{vmatrix} + \begin{vmatrix} a_{11} & a_{13} \\ a_{13} & a_{33} \end{vmatrix} + \begin{vmatrix} a_{22} & a_{23} \\ a_{23} & a_{33} \end{vmatrix}, I_3 = \begin{vmatrix} a_{11} & a_{12} & a_{13} \\ a_{12} & a_{22} & a_{23} \\ a_{13} & a_{23} & a_{33} \end{vmatrix}, \text{ and}$$

$$I_4 = \begin{vmatrix} a_{11} & a_{12} & a_{13} & a_{14} \\ a_{12} & a_{22} & a_{23} & a_{24} \\ a_{13} & a_{23} & a_{33} & a_{34} \\ a_{14} & a_{24} & a_{34} & a_{44} \end{vmatrix}.$$

**Step 2.** We find the central point  $(x_0, y_0, z_0)$  of  $S$  which satisfies:

$$\begin{cases} a_{11}x_0 + a_{12}y_0 + a_{13}z_0 + a_{14} = 0 \\ a_{12}x_0 + a_{22}y_0 + a_{23}z_0 + a_{24} = 0 \\ a_{13}x_0 + a_{23}y_0 + a_{33}z_0 + a_{34} = 0 \end{cases} \quad (5)$$

**Step 3.** We establish the characteristic equation of  $S$ :  $-\lambda^3 + I_1\lambda^2 - I_2\lambda + I_3 = 0$ . (6)

And find the characteristic roots  $\lambda_1, \lambda_2, \lambda_3$ .

**Step 4.** We substitute  $\lambda_i (i = 1, 2, 3)$  into the system of equations,

$$\begin{cases} (a_{11} - \lambda_i)X + a_{12}Y + a_{13}Z = 0 \\ a_{12}X + (a_{22} - \lambda_i)Y + a_{23}Z = 0, \\ a_{13}X + a_{23}Y + (a_{33} - \lambda_i)Z = 0 \end{cases} \quad (7)$$

and obtain the principal directions  $X_i : Y_i : Z_i (i = 1, 2, 3)$  of  $S$ .

**Step 5.** We substitute  $X_i, Y_i, Z_i (i = 1, 2, 3)$  into the equation,  $X_i(a_{11}x + a_{12}y + a_{13}z + a_{14}) + Y_i(a_{12}x + a_{22}y + a_{23}z + a_{24}) + Z_i(a_{13}x + a_{23}y + a_{33}z + a_{34}) = 0$ , and get the principal diametral planes of  $S$ .

**Step 6.** We use three orthogonal principal diametral planes as new coordinate planes, and make an orthogonal transformation, so that  $S$  is transformed into  $S'$  of the form  $\frac{x^2}{a^2} + \frac{y^2}{b^2} + \frac{z^2}{c^2} = 1$  and  $P$  is transformed into a plane  $P_1$ . (If we get only two principal diametral planes, we can use the central point  $(x_0, y_0, z_0)$  to get the third one.)

**Step 7.** Through a linear transformation:  $x = ax', y = by', z = cz'$ ,  $S'$  is transformed into an unit sphere  $S''$ , and the plane  $P_1$  is transformed into  $P_2$ :  $a_2x' + b_2y' + c_2z' = d_2$ . We compute the Jacobian determinant  $J$  of the linear transformation.

**Step 8.** We compute the volumes of spherical segments determined by the formula  $\pi h^2(R-h/3)$ , and get the volumes of pieces bounded by  $S$  and  $P$  with the formula  $\pi h^2(R-h/3)|J|$ , where  $h$  is the height of the corresponding spherical segment.

**Theorem 3.10** Let  $S$  be a real closed quadric of the form (4), and the plane  $P: ax + by + cz = d$  cut the solid bounded by  $S$  into only two pieces. Then the volumes of the two pieces can be computed by above steps 1 through 8.

We call above method as **the linear transformation method**, the detailed process of calculations is given by the procedure (named as “**quadvol**” see [14]). Readers need input the parameters (fun and pla) of “**quadvol**” and the variables given by “x, y, z”. For example, one may write fun= $x^2+y^2/4+z^2/9-1$ . We show in the following Example 3.11 how the procedure, “**quadvol**” is implemented.

**Example 3.11** Let  $S$  be the surface  $14x^2 + 27y^2 + 98z^2 + 4yz - 34xz - 8xy + 28x + 20y - 74z - 169 = 0$  and  $P$  be the plane  $x + 3y - 2z = 1$  ( $S$  and  $P$  are same as in example 3.8). Find the respective volumes bounded by  $S$  and  $P$ .

**Solution:** Set up the expressions for  $S$  and  $P$ , then run the procedure “**quadvol**”, we have following results.

```
>fun3:=14*x^2+27*y^2+98*z^2+4*y*z-34*x*z-8*x*y+28*x+20*y-74*z-169:
>pla:=x+3*y-2*z-1:
>quadvol(fun3, pla);
    “the volume of one piece bounded by the plane P and the surface S is 14.906196 or”
    4.744789434π
    “the volume of the other piece bounded by the plane P and the surface S is 53.394332 or”
    16.99594358π
```

**Example 3.12** Let  $S$  be the surface  $x^2 + \frac{y^2}{9} + \frac{(z-3)^2}{16} - 1 = 0$ , and  $P$  be the plane  $3x - 2y - z = -1$  ( $S$  and  $P$  are same as in example 3.9). Find the respective volumes bounded by  $S$  and the planes.

**Solution:** Set up the expressions for  $S$  and  $P$ , then run the procedure “**quadvol**”, we get following results.

```
>fun4:=x^2+y^2/9+(z-3)^2/16-1:
>pla:=3*x-2*y-z+1:
>quadvol(fun4, pla);
    “the volume of one piece bounded by the plane P and the surface S is 15.689999 or”
    4.994281769π
    “the volume of the other piece bounded by the plane P and the surface S is 34.575484 or”
    11.00571824π
```

Comparing with the results in Example 3.9, we see that the answer obtained through the integral method is very close to that of method of the linear transformation. However, integral method produces a relative large error when some cases are considered; as we have seen in Example 3.9 that if we use planes  $P_1: 5x-3y-z+12=0$  or  $P_2: 3x-y+z+1=0$ , we know that the cases were

computed incorrectly by the integral method if the parameters in the procedure “**parvol**” were not chosen appropriately. We use the linear transformation method to calculate as follows.

>pla1:=5\*x-3\*y-z+12:

>quadvol(fun4, pla1);

“the volume of one piece bounded by the plane P and the surface S is 1.212944 or”  
0.3860921778 $\pi$

“the volume of the other piece bounded by the plane P and the surface S is 49.052538 or”  
15.61390782 $\pi$

>pla2:=3\*x-y+z+1:

>quadvol(fun4, pla2);

“the volume of one piece bounded by the plane P and the surface S is 3.328051 or”  
1.059351378 $\pi$

“the volume of the other piece bounded by the plane P and the surface S is 46.937432 or”  
14.94064862 $\pi$

**Remark:** When you run the procedure “**quadvol**”, a dialogue box may occur to ask you input the values of X1,Y1, Z1, according to the free variable, such as, [X1=X1,Y1=Y1, Z1=0] or [X1=X1, Y1=Y1, Z1=Z1]. In the first case, X1,Y1 are free variables, we can take any non-zero vector [X1,Y1, 0] as the values of X1,Y1, for example, X1=1,Y1=0 or X1=1,Y1=2. In the second case, X1, Y1, Z1 are free variables, we can take any two non-zero orthogonal vectors [X1, Y1, Z1] and [X2, Y2, Z2] as principle directions.

#### 4. Summary

In this paper, we programmed Maple procedures for finding the signed area bounded by a parametric curve with respect to a slanted line and the signed volume bounded by a smooth parametric surface with respect to a plane. The procedures are based on the results in [1]. We provided some examples to illustrate the process of computing. We should point out that the formula (1) or (2) can only be used to compute net signed areas or net signed volumes of regions, it may not be feasible to determine the real areas or volumes of regions due to the interactions of the curves and the lines or the interactions of the surfaces and the planes; but it is feasible to do so for many problems of interest. In section 3, two methods (integral and linear transformation) of finding the volume of the solid bounded by an orientable smooth surface and a plane are discussed, for the linear transformation method, a procedure is presented that computes the volumes of an ellipsoid crossed by a plane in all cases. Some examples are given to explain when the two different methods will get the same answers, and to show the advantages and disadvantages of the two methods.

## References

- [1] Yang, Wei-Chi and Lo, Min-Lin, Finding Signed Areas and Volumes Inspired by Technology, The Electronic Journal of Mathematics and Technology (eJMT), ISSN 1933-2823, Issue 2, Volume 2, pp. 133-150, 2008.
- [2] Martha L. Abell and James P. Braselton, Maple by Examples (Third Edition), ELSEVIER Academic Press, ISBN: 0-12-088526-3, 2005.
- [3] Stewart, J., Calculus Concepts & Contexts, Brooks Cole (Fourth Edition), ISBN-10: 0-495-55742-0, ISBN-13: 978-0-495-55742-5, 2009.
- [4] Lingen Lv and Zidao Xu, Analytic Geometry (Fourth Edition, in Chinese), Higher Education Press, ISBN 7-04-019364-7, 2006.

## Software Packages and Supplemental Electronic Materials

- [5] A product of Maplesoft, <http://www.maplesoft.com/>.
- [6] Xia, S.X., Yang, W.-C., and Shelomovskiy V. V., [Maple worksheet for Example 2.3.](#)
- [7] Xia, S.X., Yang, W.-C., and Shelomovskiy V. V., [Maple worksheet for Example 2.4.](#)
- [8] Xia, S.X., Yang, W.-C., and Shelomovskiy V. V., [Maple worksheet for Example 3.3.](#)
- [9] Xia, S.X., Yang, W.-C., and Shelomovskiy V. V., [Maple worksheet for Example 3.4.](#)
- [10] Xia, S.X., Yang, W.-C., and Shelomovskiy V. V., [Maple worksheet for Example 3.6.](#)
- [11] Xia, S.X., Yang, W.-C., and Shelomovskiy V. V., [Maple worksheet for Example 3.7.](#)
- [12] Xia, S.X., Yang, W.-C., and Shelomovskiy V. V., [Maple worksheet for Example 3.8.](#)
- [13] Xia, S.X., Yang, W.-C., and Shelomovskiy V. V., [Maple worksheet for Example 3.9.](#)
- [14] Xia, S.X., Yang, W.-C., and Shelomovskiy V. V., [Maple worksheet for Example 3.11.](#)
- [15] Xia, S.X., Yang, W.-C., and Shelomovskiy V. V., [Maple worksheet for Example 3.12.](#)